@echo off
setlocal

REM 删除 build 和 hf_build 目录（如果存在）
echo 正在删除 build 、hf_build 、modelscope_build 目录...
if exist build (
    rmdir /s /q build
    echo 删除了 build 目录。
)
if exist hf_build (
    rmdir /s /q hf_build
    echo 删除了 hf_build 目录。
)
if exist modelscope_build (
    rmdir /s /q modelscope_build
    echo 删除了 modelscope_build 目录。
)


REM 执行 hf_setup.py build
echo 正在执行 hf_setup.py build...
call python hf_setup.py build 1>nul
if %ERRORLEVEL% neq 0 (
    echo hf_setup.py build 失败，退出脚本。
    exit /b 1
)

REM 将 build 文件夹重命名为 hf_build
echo 正在将build文件夹重命名为hf_build...
ren build hf_build
if %ERRORLEVEL% neq 0 (
    echo 重命名失败，退出脚本。
    exit /b 1
)

REM 执行 modelscope_setup.py build
echo 正在执行 modelscope_setup.py build...
call python modelscope_setup.py build 1>nul
if %ERRORLEVEL% neq 0 (
    echo modelscope_setup.py build 失败，退出脚本。
    exit /b 1
)


REM 将 build 文件夹重命名为 modelscope_build
echo 正在将build文件夹重命名为modelscope_build...
ren build modelscope_build
if %ERRORLEVEL% neq 0 (
    echo 重命名失败，退出脚本。
    exit /b 1
)

if exist "hf_build\exe.win-amd64-3.12\frozen_application_license.txt" (
    REM 删除文件
    del /f /q "hf_build\exe.win-amd64-3.12\frozen_application_license.txt"
    echo 文件 hf_build\exe.win-amd64-3.12\frozen_application_license.txt 已被删除。
)
if exist "modelscope_build\exe.win-amd64-3.12\frozen_application_license.txt" (
    REM 删除文件
    del /f /q "modelscope_build\exe.win-amd64-3.12\frozen_application_license.txt"
    echo 文件 modelscope_build\exe.win-amd64-3.12\frozen_application_license.txt 已被删除。
)

echo ""
echo 所有 Build 操作已完成，请执行 package.bat 将文件夹打包为zip。


endlocal
pause
